
Reason for change:
(

Following application of CR R5-113662, on inter-system change from S1 mode to A/Gb mode, a PDP context corresponding to the default EPS bearer will be activated. This may cause a UE to initiate the LLC XID negotiation procedure. Current implementation of TTCN contains functions to handle the XID signalling, but these are not currently called from the existing GERAN PTC functions.




Summary of change:

Update GERAN PTC functions to handle XID procedure following a ROUTING AREA UPDATE on intersystem change from S1 mode. 

Change 1

Testcase name
All testcases involving EUTRAN->GERAN intersystem change

Protocol
Idle mode

Function Name
f_GERAN_RAU()

Reason for change
Allow XID procedure during RAU 

Summary of change
Add new parameter p_WaitForXID and pass through to f_GERAN_RAUWithoutRAUReq()

Source of change
GERAN_CommonFunctions.ttcn

Before:
    //----------------------------------------------------------------------------

  /*

   * @desc      To perform the Routing Area Update procedure

   * @param     p_CellId

   * @param     p_FOR

   * @param     p_Type

   * @param     p_DoAuth            (default value: false)

   * @status    APPROVED

   */  

function f_GERAN_RAU(GERAN_CellId_Type p_CellId,

                       template (present) B1_Type p_FOR,

                       template (present) B3_Type p_Type,

                       boolean p_DoAuth := false) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    // Receive RAU Request Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND ( cr_G_RA_UpdReqAny (cr_GMM_UpdateType( p_FOR, p_Type), cr_RAI (omit), cr_PTMSI_SignatureAny ifpresent, ?)));

    v_UpdateType := v_LLCDataInd.unit.msg.rauReq.updateType.updateValue;

    if (v_UpdateType != '000'B){

      v_UpdateType := '001'B;

    }

    f_GERAN_RAUWithoutRAUReq (p_CellId, v_UpdateType, p_DoAuth);

  }// end of f_GERAN_RAU

After:
    //----------------------------------------------------------------------------

  /*

   * @desc      To perform the Routing Area Update procedure

   * @param     p_CellId

   * @param     p_FOR

   * @param     p_Type

   * @param     p_DoAuth            (default value: false)

   * @param     p_WaitForXID        (default value: false)  set to true if UE may send an XID 

   * @status    APPROVED

   */  

function f_GERAN_RAU(GERAN_CellId_Type p_CellId,

                       template (present) B1_Type p_FOR,

                       template (present) B3_Type p_Type,

                       boolean p_DoAuth := false,

                       boolean p_WaitForXID := false) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    // Receive RAU Request Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND ( cr_G_RA_UpdReqAny (cr_GMM_UpdateType( p_FOR, p_Type), cr_RAI (omit), cr_PTMSI_SignatureAny ifpresent, ?)));

    v_UpdateType := v_LLCDataInd.unit.msg.rauReq.updateType.updateValue;

    if (v_UpdateType != '000'B){

      v_UpdateType := '001'B;

    }

    f_GERAN_RAUWithoutRAUReq (p_CellId, v_UpdateType, p_DoAuth, p_WaitForXID);

  }// end of f_GERAN_RAU

Change 2

Testcase name
All testcases involving EUTRAN->GERAN intersystem change

Protocol
Idle mode

Function Name
f_GERAN_RAUWithoutRAUReq()

Reason for change
Allow XID procedure during RAU 

Summary of change
Add new parameter p_WaitForXID. If this is set to TRUE, call function f_LLC_XID() following sending of RAU ACCEPT.

Source of change
GERAN_CommonFunctions.ttcn

Before:
  //----------------------------------------------------------------------------

  /*

   * @desc      To perform the Routing Area Update Procedure, after the RAU Request message has been received

   * @param     p_CellId

   * @param     p_UpdateType

   * @param     p_DoAuth            (default value: false)

   * @status    APPROVED

   */  

function f_GERAN_RAUWithoutRAUReq (GERAN_CellId_Type p_CellId,

                                     B3_Type p_UpdateType,

                                     boolean p_DoAuth := false) runs on GERAN_PTC

  {

    …..

    // Send RAU Accept

    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(p_CellId),

                                       tsc_LLCSapi_GMM,

                                       tsc_LLC_PM,

                                       px_GSM_CipheringOnOff,

                                       cs_G_RA_UpdAcc3 (cs_GMM_UpdateResult( p_UpdateType ),

                                                        cs_RAI(omit,

                                                               cs_LAI(omit, v_PLMN, v_LAC),

                                                               v_RAC),

                                                        cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                        cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def),

                                                        omit )));

    //Receive RAU Complete Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND (cr_G_RA_UpdComplete));

  } // end of f_GERAN_RAUWithoutRAUReq

After:
  /*

   * @desc      To perform the Routing Area Update Procedure, after the RAU Request message has been received

   * @param     p_CellId

   * @param     p_UpdateType

   * @param     p_DoAuth            (default value: false)

   * @param     p_WaitForXID        (default value: false)  set to true if UE may send an XID 

   * @status    APPROVED

   */

  function f_GERAN_RAUWithoutRAUReq (GERAN_CellId_Type p_CellId,

                                     B3_Type p_UpdateType,

                                     boolean p_DoAuth := false,

                                     boolean p_WaitForXID := false) runs on GERAN_PTC

  {

….

    // Send RAU Accept

    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(p_CellId),

                                       tsc_LLCSapi_GMM,

                                       tsc_LLC_PM,

                                       px_GSM_CipheringOnOff,

                                       cs_G_RA_UpdAcc3 (cs_GMM_UpdateResult( p_UpdateType ),

                                                        cs_RAI(omit,

                                                               cs_LAI(omit, v_PLMN, v_LAC),

                                                               v_RAC),

                                                        cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                        cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def),

                                                        omit )));

    if (p_WaitForXID) { // 

      f_LLC_XID (p_CellId, p_WaitForXID);

    } else {
      //Receive RAU Complete Message

      v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND (cr_G_RA_UpdComplete));

    }
  } // end of f_GERAN_RAUWithoutRAUReq

Change 3

Testcase name
All testcases involving EUTRAN->GERAN intersystem change

Protocol
Idle mode

Function Name
f_GERAN_508CheckCampOnNewGeranCell()

Reason for change
Allow XID procedure during RAU 

Summary of change
Add new parameter p_WaitForXID and pass through to f_GERAN_RAU()

Source of change
GERAN_CommonFunctions.ttcn

Before:
    //----------------------------------------------------------------------------

  /*

   * @desc       Procedure which aims at checking whether the UE is camping on a new GERAN cell of a test case or not.

   *             The procedure is defined in 36.508 table 6.4.2.9-1.

   * @param     p_CellId

   * @status    APPROVED

   */

  function f_GERAN_508CheckCampOnNewGeranCell ( GERAN_CellId_Type p_CellId ) runs on GERAN_PTC

  {

    f_GERAN_RAU (p_CellId, '0'B, (tsc_I_UpdateTypeRAOnly, tsc_I_UpdateTypeCombined), true); // @sic R5s110087 sic@

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "");

  }

After:
    //----------------------------------------------------------------------------

  /*

   * @desc       Procedure which aims at checking whether the UE is camping on a new GERAN cell of a test case or not.

   *             The procedure is defined in 36.508 table 6.4.2.9-1.

   * @param     p_CellId

   * @param     p_WaitForXID (default: true) set this to true if the UE may send an XID after the RAU    

   * @status    APPROVED

   */

  function f_GERAN_508CheckCampOnNewGeranCell ( GERAN_CellId_Type p_CellId, boolean p_WaitForXID := true) runs on GERAN_PTC

  {

    f_GERAN_RAU (p_CellId, '0'B, (tsc_I_UpdateTypeRAOnly, tsc_I_UpdateTypeCombined), true, p_WaitForXID); 

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "");

  }

Change 4

Testcase name
All testcases involving EUTRAN->GERAN intersystem change

Protocol
Idle mode

Function Name
f_GERAN_IdleMode_RAU()

Reason for change
Allow XID procedure during RAU 

Summary of change
Set new parameter p_WaitForXID to TRUE when calling f_GERAN_508CheckCampOnNewCell()

Source of change
GERAN_CommonFunctions.ttcn

Before:
  //----------------------------------------------------------------------------

  /*

   * @desc      Function used in Idle Mode test cases to perform Routing Area Update on the new GERAN cell.

   * @param     p_CellId   ..  The GERAN cell on which the random access request and the RAU will happen

   * @status    APPROVED

   */

  function f_GERAN_IdleMode_RAU(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    const float const_IdleMode_GenericTimer_GERAN := 360.0;   /* Idle mode generic timer according to 36.523-3 cl. 7.8 */

    timer t_IdleMode_GenericTimer := f_UG_SetTimerToleranceMax(const_IdleMode_GenericTimer_GERAN);

    t_IdleMode_GenericTimer.start;

    f_GERAN_508CheckCampOnNewGeranCell ( p_CellId );

    t_IdleMode_GenericTimer.stop;

  }

After:
  //----------------------------------------------------------------------------

  /*

   * @desc      Function used in Idle Mode test cases to perform Routing Area Update on the new GERAN cell.

   * @param     p_CellId   ..  The GERAN cell on which the random access request and the RAU will happen

   * @status    APPROVED

   */

  function f_GERAN_IdleMode_RAU(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    const float const_IdleMode_GenericTimer_GERAN := 360.0;   /* Idle mode generic timer according to 36.523-3 cl. 7.8 */

    timer t_IdleMode_GenericTimer := f_UG_SetTimerToleranceMax(const_IdleMode_GenericTimer_GERAN);

    t_IdleMode_GenericTimer.start;

    f_GERAN_508CheckCampOnNewGeranCell ( p_CellId, true );

    t_IdleMode_GenericTimer.stop;

  }

Change 5

Testcase name
All testcases involving EUTRAN->GERAN intersystem change

Protocol
Idle mode

Function Name
f_LLC_XID()

Reason for change
Range check on access cause is incorrect 

Summary of change
Change “<” to “>” in range check on access cause

Source of change
GERAN_CommonFunctions.ttcn

Before:
  function f_LLC_XID (GERAN_CellId_Type p_CellId, boolean p_RAUComplete) runs on GERAN_PTC

  {

    ………….

    [] G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId , tsc_PhyCh0 , tsc_RACH , ? , ? , cr_ChanReqOnePhase ))

        -> value v_ChanReq

      {

        v_RFN := v_ChanReq.access.rfn;

        v_RR_RA := bit2int(v_ChanReq.access.burst.chReq.estCauRandomRef);

        if ((119 < v_RR_RA) or (v_RR_RA > 127)) {// establishment cause should be one phase access

          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "wrong establishment cause");

        }

        fl_ULTBFOnePhase (p_CellId, v_RFN, v_RR_RA);

        repeat;

      }

    []  t_Wait10S.timeout {} // No XID coming

    }

  } //end of f_LLC_XID

After:
function f_LLC_XID (GERAN_CellId_Type p_CellId, boolean p_RAUComplete) runs on GERAN_PTC

  {

    ………….

    [] G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId , tsc_PhyCh0 , tsc_RACH , ? , ? , cr_ChanReqOnePhase ))

        -> value v_ChanReq

      {

        v_RFN := v_ChanReq.access.rfn;

        v_RR_RA := bit2int(v_ChanReq.access.burst.chReq.estCauRandomRef);

        if ((119 > v_RR_RA) or (v_RR_RA > 127)) {// establishment cause should be one phase access

          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "wrong establishment cause");

        }

        fl_ULTBFOnePhase (p_CellId, v_RFN, v_RR_RA);

        repeat;

      }

    []  t_Wait10S.timeout {} // No XID coming

    }

  } //end of f_LLC_XID

