
1. Overview

This document lists changes needed to correct problems in the TTCN implementation of test case 6.2.1.3  which is part of the LTE/SAE test suite ‘iwd-EUTRA-B2011-03_D11wk37’. 

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
TC_6_2_1_3

Test Group:
InterRat_PLMNSelection.ttcn

ATS Version:
iwd-EUTRA-B2011-03_D11wk37
System Simulator used:
R&S CMW500 
UE used:
Qualcomm MDM 9200

Verification Status:
PASS


4. Corrections required for test case 6.2.1.3
This section describes the TTCN changes required to make test case 6.2.1.3 run correctly with a LTE UE. The TTCN used to pass this test case belongs to ‘iwd-EUTRA-B2011-03_D11wk37’ release.

Change 1 – Correction to ‘f_TC_6_2_1_3_EUTRA ()’

Function name
f_TC_6_2_1_3_EUTRA()

Reason for change
1. PLMN2 value for cell 1 is not as per TS 36.523-1 table 6.0.1-1. Moreover, the table number in the comments line above is pointing to a different TC. 

PLMN index in function f_EUTRA_CellInfo_SetPLMN_2Entries should be different from the default value (= ‘0’) as the UE shall select PLMN2 instead of PLMN13 which has lower prirority in EUTRA. This index is used to update the selected PLMN identity on the stored GUTI parameters. The updated value shall be used later to compute the security keys during the registration procedure. And as the UE must select PLMN2 during registration, index ‘1’ shall be stored which corresponds to PLMN2.

2. As eutra_cell1 is configured with a list of PLMNs containing 2 elements (PLMN13, PLMN2), and the second PLMN identity (= PLMN2) having higher priority than the PLMN on position 1 (= PLMN13), the UE shall report a selected PLMN value of ‘2’ via function f_EUTRA_Preamble_Reject. 

Summary of change
1. PLMN2 value modified from ‘cs_HPLMN_001_21’ to ‘cs_HPLMN_002_11’ as per table 6.0.1-1. Referred TC number corrected in the comments line.

GUti_params updated to PLMN2 (Index =1).
2. Function f_EUTRA_Preamble_Reject has been modified to include a new parameter which indicates the PLMN identity that was selected by the UE. 

TTCN module
InterRat_PLMNSelection.ttcn

MCC160 Comment


Before change

  function f_TC_6_2_1_3_EUTRA ( ) runs on EUTRA_PTC {
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var SRB_COMMON_IND v_ReceivedAsp;
    var template (value) CellPowerList_Type v_CellPowerList_AtT0;
    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(NORMAL); // @sic R5s110176 sic@
    //var template (present) SecurityHeaderType v_AttReqSecurityHeader := (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected);
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var B3_Type v_PDNType := f_GetPdnType();
    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;
    v_CellPowerList_AtT0 := {
      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),
      cs_CellPower (eutra_Cell12, tsc_ServingCellRS_EPRE)
    };
    //Initialise all cell, security and mobile parameters
    f_EUTRA_Init (c4);//SIB2, SIB3 & SIB6 are transmitted
    //Set TAC according to 36.523-1 Table 6.0.1-2
    //TAC for Cell 1 is set to 1 in f_EUTRA_Init ();
    f_EUTRA_CellInfo_SetTAC (eutra_Cell12, tsc_IdleMode_TAC_Cell12);
    //Set PLMN according to Table 6.2.1.2.3.2-1: Cell PLMN identities
    f_EUTRA_CellInfo_SetPLMN_2Entries (eutra_Cell1, cs_HPLMN_002_81, cs_HPLMN_001_21);
    f_EUTRA_CellInfo_SetPLMN_2Entries (eutra_Cell12, cs_HPLMN_002_81, cs_HPLMN_002_91);
    //Set Minimum UL flag for Cell 1
    f_EUTRA_SetCellConfigCapability ( eutra_Cell12, minimumUplinkCell );
    //Create and configure all cells
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_EUTRA_CellConfig_Def (eutra_Cell12);
    //Equip UE with a USIM containing default values except for those shown in Table 6.2.1.3.3.1-1
    f_UT_USIM_Insert (UT, "36.523-1 Table 6.2.1.3.3.1-1");
    //Preamble to remove RPLMN in UE and bring UE to initial state
    f_EUTRA_Preamble_Reject(eutra_Cell1, tsc_EmmCause_IllegalMe, NORMAL);
    //Set initial cell power level according to T0
    f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger);
    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT0);
    //Wait for correct power levels and then start the test case
    f_WaitForIRATCoOrd_Trigger ( UTRAN );
    f_EUTRA_TestBody_Set ( true );
    

After change

  function f_TC_6_2_1_3_EUTRA ( ) runs on EUTRA_PTC {
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var SRB_COMMON_IND v_ReceivedAsp;
    var template (value) CellPowerList_Type v_CellPowerList_AtT0;
    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(NORMAL); // @sic R5s110176 sic@
    //var template (present) SecurityHeaderType v_AttReqSecurityHeader := (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected);
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var B3_Type v_PDNType := f_GetPdnType();
    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;
    v_CellPowerList_AtT0 := {
      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),
      cs_CellPower (eutra_Cell12, tsc_ServingCellRS_EPRE)
    };
    //Initialise all cell, security and mobile parameters
    f_EUTRA_Init (c4);//SIB2, SIB3 & SIB6 are transmitted
    //Set TAC according to 36.523-1 Table 6.0.1-2
    //TAC for Cell 1 is set to 1 in f_EUTRA_Init ();
    f_EUTRA_CellInfo_SetTAC (eutra_Cell12, tsc_IdleMode_TAC_Cell12);
    //Set PLMN according to Table 6.2.1.3.3.2-1: Cell PLMN identities
    f_EUTRA_CellInfo_SetPLMN_2Entries (eutra_Cell1, cs_HPLMN_002_81, cs_HPLMN_002_11, 1);
f_EUTRA_CellInfo_SetPLMN_2Entries (eutra_Cell12, cs_HPLMN_002_81, cs_HPLMN_002_91);

    //Set Minimum UL flag for Cell 1
    f_EUTRA_SetCellConfigCapability ( eutra_Cell12, minimumUplinkCell );
    //Create and configure all cells
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_EUTRA_CellConfig_Def (eutra_Cell12);
    //Equip UE with a USIM containing default values except for those shown in Table 6.2.1.3.3.1-1
    f_UT_USIM_Insert (UT, "36.523-1 Table 6.2.1.3.3.1-1");
    //Preamble to remove RPLMN in UE and bring UE to initial state
    f_EUTRA_Preamble_Reject(eutra_Cell1, tsc_EmmCause_IllegalMe, NORMAL, 2);  
    //Set initial cell power level according to T0
    f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger);
    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT0);
    //Wait for correct power levels and then start the test case
    f_WaitForIRATCoOrd_Trigger ( UTRAN );
    f_EUTRA_TestBody_Set ( true );
    

Change 2 – Correction to ‘f_TC_6_2_1_3_UTRAN ()’

Function name
f_TC_6_2_1_3_UTRAN ()

Reason for change
1. In order to reduce code complexity and to allow reuse of the already existing decoders for UTRAN NAS messages, it is proposed not to introduce a separate decoding structure based on a single NAS message. Especially as there is no real justification for introducting of such new structure.

2. In step 23 the Initial Direct Transfer message should be received on ‘eutra_cellDedicated’ instead of ‘eutra_Cell9’.

3. As RB3 was reconfigured to tsc_RB3_DCCH_RRC, it is required to change the expected RB ID for the Initial Direct Transfer received in step 23.

4. RB routing needs to be preconfigured before starting the RRC connection handshake in step 20 – 22. 

Summary of change
1. Decoding is done via structure RRCDataInd which already includes all NAS messages.

2. ‘eutra_Cell9’ overwritten with ‘eutra_cellDedicated’ in the template ‘car_RLC_Data_Ind’.

3. RB ID for the Initial Direct Transfer received in step 23 is changed.

4. Function f_UTRAN_SS_SetConfigRRC_RB3 has been moved just before the RRC Connection Request message in step 20.

TTCN module
InterRat_PLMNSelection_UTRAN.ttcn

MCC160 Comment


Before change

  function f_TC_6_2_1_3_UTRAN ( ) runs on UTRAN_PTC {
    var UTRAN_FDD_TDD v_FDD_TDD_Mode;
    var template (value) MasterInformationBlock v_MIB_MultiplePlmnList;
    var InitialUE_Identity v_InitialUE_Id;
    var RLC_TR_DATA_IND v_RLC_TR_DATA_IND;
    var U_RLC_AM_IND v_RLC_AM_DATA_IND;
    var integer v_T0_CPICH_Cell9, v_T0_PCCPCH_Cell9;
    var PLMN_Identity v_PLMN14;
    var PLMN_IdentityWithOptionalMCC_r6 v_PLMN13;
    var octetstring v_RauReq_Msg;
    var ROUTINGAREAUPDATEREQUEST v_ReceivedRauReq_Msg;
    var B3_Type v_UpdateValue;
    var START_Value v_START_Value;
    var template (present) EstablishmentCause v_EstCause := registration;
    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer_UTRAN;
    v_PLMN14 := {mcc := {0,0,2}, mnc := {9,1}};//PLMN14 according to 36.523-1 Table 6.0.1-1
    v_PLMN13 := {mcc := {0,0,2}, mnc := {8,1}};//PLMN13 according to 36.523-1 Table 6.0.1-1
    v_T0_CPICH_Cell9 := tsc_AttenuationServingCell;//No attnuation => serving cell
    v_T0_PCCPCH_Cell9 := tsc_AttenuationServingCell;//No attnuation => serving cell
    //Initialise all cells
    f_UTRAN_Init (EUTRA_UTRAN);
    //Get FDD or TDD mode
    v_FDD_TDD_Mode := f_UTRAN_CellInfo_GetFDD_TDD (utran_Cell9);
    //Set PLMN14 in cell 5
    f_UTRAN_CellInfo_SetPLMNIdentity (utran_Cell9, v_PLMN14);
    v_MIB_MultiplePlmnList := cds_MIB_MultiplePlmnList (v_PLMN14, {v_PLMN13});
    f_UTRAN_SysInfo_SetMIB (utran_Cell9, v_MIB_MultiplePlmnList);
    //Create and configure Cell 5
    //Cell started switched on with default power level, -60 and max attenuation
    f_UTRAN_SS_CreateCellFACH (utran_Cell9);
    //Start sending System Information
    f_UTRAN_SendDefSysInfo (utran_Cell9);
    //Wait for synch with EUTRA PTC and then set cell power level according to T0
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_UTRAN_SetCellPower(utran_Cell9, v_T0_CPICH_Cell9, v_T0_PCCPCH_Cell9);
    //Tell EUTRA PTC to start the test case
    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);
    f_UTRAN_TestBody_Set (true);
    //Receive updated Authentication parameters
    f_UTRAN_InterRAT_InitialiseAuthParams();
    //@siclog "Step 19" siclog@
    //Set cell power level according to T1
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    //@siclog "Step 20" siclog@
    //Wait for RRC CONNECTION REQUEST on cell 5
    t_IdleMode_GenericTimer.start;
    U_TM.receive (car_RRC_ConnReq (utran_Cell9,
                                   tsc_RB0,
                                   cr_108_RRC_ConnReq_r8 (v_EstCause))) -> value v_RLC_TR_DATA_IND;
    t_IdleMode_GenericTimer.stop; //@sic R5s110633 : Additional changes sic@                                   
    v_InitialUE_Id := v_RLC_TR_DATA_IND.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
    //@siclog "Step 21 - 22" siclog@
    //Send RRC CONNECTION SETUP and receive RRC CONNECTION SETUP COMPLETE
    f_UTRAN_RRC_ConnEst_WithoutRRCConnectionReq (utran_Cell9, v_InitialUE_Id);
    f_UTRAN_SS_SetConfigRRC_RB3 (utran_Cell9);
    //@siclog "Step 23" siclog@
    //Receive Routing Area Request message with PLMN identity set to PLMN14
    U_AM.receive (car_RLC_Data_Ind (utran_Cell9,
                                    tsc_RB3,
                                    cr_InitDirectTransfer (?,
                                                           v_PLMN14))) -> value v_RLC_AM_DATA_IND;
    //Decode the RAU Req message
    v_RauReq_Msg := v_RLC_AM_DATA_IND.data.aM_message.uL_DCCH_Message.message_.initialDirectTransfer.nas_Message;
    v_ReceivedRauReq_Msg := f_DecodeRauReq_Msg (v_RauReq_Msg);
    v_UpdateValue := v_ReceivedRauReq_Msg.updateType.updateValue;
    v_START_Value := v_RLC_AM_DATA_IND.data.aM_message.uL_DCCH_Message.message_.initialDirectTransfer.v3a0NonCriticalExtensions.initialDirectTransfer_v3a0ext.start_Value;
    f_UTRAN_SS_RemoveConfigRRC_RB3 (utran_Cell9);
    //@siclog "Step 24 - 27" siclog@
    //Perform 2-10 of the Routing Area Update procedure (36.508 cl. 6.4.2.8)
    f_UTRAN_GMM_RAU_Without_RAUReq (utran_Cell9, v_START_Value, v_UpdateValue, false);
    //@siclog "Step 28 - 29" siclog@
    //Release RRC Connection
    f_UTRAN_RRC_ConnRel (utran_Cell9, cell_Fach_Dcch);
    //@siclog "Step 30" siclog@
    //Check if PLMN14 is indicated by UE
    f_UT_PLMN_Check (UT, "00291");
    //* @verdict pass PLMN14 is indicated by UE
    f_UTRAN_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.2.1.3 Step 32");
    f_UTRAN_TestBody_Set (false);
    //Tell the EUTRA component that the test case is over
    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);
    //Switch/Power off UE and release cell
    f_UTRAN_Postamble (utran_Cell9, U1_IDLE);
  }
}

After change

  function f_TC_6_2_1_3_UTRAN ( ) runs on UTRAN_PTC {
    var UTRAN_FDD_TDD v_FDD_TDD_Mode;
    var template (value) MasterInformationBlock v_MIB_MultiplePlmnList;
    var InitialUE_Identity v_InitialUE_Id;
    var RLC_TR_DATA_IND v_RLC_TR_DATA_IND;
    var U_RLC_AM_IND v_RLC_AM_DATA_IND;
    var integer v_T0_CPICH_Cell9, v_T0_PCCPCH_Cell9;
    var PLMN_Identity v_PLMN14;
    var PLMN_IdentityWithOptionalMCC_r6 v_PLMN13;
    var octetstring v_RauReq_Msg;
    var RRCDataIndType v_ReceivedRauReq_Msg;
    var B3_Type v_UpdateValue;
    var START_Value v_START_Value;
    var template (present) EstablishmentCause v_EstCause := registration;
    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer_UTRAN;
    v_PLMN14 := {mcc := {0,0,2}, mnc := {9,1}};//PLMN14 according to 36.523-1 Table 6.0.1-1
    v_PLMN13 := {mcc := {0,0,2}, mnc := {8,1}};//PLMN13 according to 36.523-1 Table 6.0.1-1
    v_T0_CPICH_Cell9 := tsc_AttenuationServingCell;//No attnuation => serving cell
    v_T0_PCCPCH_Cell9 := tsc_AttenuationServingCell;//No attnuation => serving cell
    //Initialise all cells
    f_UTRAN_Init (EUTRA_UTRAN);
    //Get FDD or TDD mode
    v_FDD_TDD_Mode := f_UTRAN_CellInfo_GetFDD_TDD (utran_Cell9);
    //Set PLMN14 in cell 5
    f_UTRAN_CellInfo_SetPLMNIdentity (utran_Cell9, v_PLMN14);
    v_MIB_MultiplePlmnList := cds_MIB_MultiplePlmnList (v_PLMN14, {v_PLMN13});
    f_UTRAN_SysInfo_SetMIB (utran_Cell9, v_MIB_MultiplePlmnList);
    //Create and configure Cell 5
    //Cell started switched on with default power level, -60 and max attenuation
    f_UTRAN_SS_CreateCellFACH (utran_Cell9);
    //Start sending System Information
    f_UTRAN_SendDefSysInfo (utran_Cell9);
    //Wait for synch with EUTRA PTC and then set cell power level according to T0
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_UTRAN_SetCellPower(utran_Cell9, v_T0_CPICH_Cell9, v_T0_PCCPCH_Cell9);
    //Tell EUTRA PTC to start the test case
    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);
    f_UTRAN_TestBody_Set (true);
    //Receive updated Authentication parameters
    f_UTRAN_InterRAT_InitialiseAuthParams();
    //@siclog "Step 19" siclog@
    //Set cell power level according to T1
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    //@siclog "Step 20" siclog@
    //Wait for RRC CONNECTION REQUEST on cell 5
    t_IdleMode_GenericTimer.start;
    f_UTRAN_SS_SetConfigRRC_RB3 (utran_Cell9);
    U_TM.receive (car_RRC_ConnReq (utran_Cell9,
                                   tsc_RB0,
                                   cr_108_RRC_ConnReq_r8 (v_EstCause))) -> value v_RLC_TR_DATA_IND;
    t_IdleMode_GenericTimer.stop; //@sic R5s110633 : Additional changes sic@                                   
    v_InitialUE_Id := v_RLC_TR_DATA_IND.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
    //@siclog "Step 21 - 22" siclog@
    //Send RRC CONNECTION SETUP and receive RRC CONNECTION SETUP COMPLETE
    f_UTRAN_RRC_ConnEst_WithoutRRCConnectionReq (utran_Cell9, v_InitialUE_Id);
    //@siclog "Step 23" siclog@
    //Receive Routing Area Request message with PLMN identity set to PLMN14
    U_AM.receive (car_RLC_Data_Ind (utran_CellDedicated, 
                                   tsc_RB3_DCCH_RRC,
                                    cr_InitDirectTransfer (?,
                                                           v_PLMN14))) -> value v_RLC_AM_DATA_IND;
    //Decode the RAU Req message
    v_RauReq_Msg := v_RLC_AM_DATA_IND.data.aM_message.uL_DCCH_Message.message_.initialDirectTransfer.nas_Message;
    v_ReceivedRauReq_Msg := f_DecodeRrcDataInd_Msg (v_RauReq_Msg);
v_UpdateValue := v_ReceivedRauReq_Msg.routingAreaUpdateRequest.updateType.updateValue;  

    v_START_Value := v_RLC_AM_DATA_IND.data.aM_message.uL_DCCH_Message.message_.initialDirectTransfer.v3a0NonCriticalExtensions.initialDirectTransfer_v3a0ext.start_Value;
    f_UTRAN_SS_RemoveConfigRRC_RB3 (utran_Cell9);
    //@siclog "Step 24 - 27" siclog@
    //Perform 2-10 of the Routing Area Update procedure (36.508 cl. 6.4.2.8)
    f_UTRAN_GMM_RAU_Without_RAUReq (utran_Cell9, v_START_Value, v_UpdateValue, false);
    //@siclog "Step 28 - 29" siclog@
    //Release RRC Connection
    f_UTRAN_RRC_ConnRel (utran_Cell9, cell_Fach_Dcch);
    //@siclog "Step 30" siclog@
    //Check if PLMN14 is indicated by UE
    f_UT_PLMN_Check (UT, "00291");
    //* @verdict pass PLMN14 is indicated by UE
    f_UTRAN_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.2.1.3 Step 32");
    f_UTRAN_TestBody_Set (false);
    //Tell the EUTRA component that the test case is over
    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);
    //Switch/Power off UE and release cell
    f_UTRAN_Postamble (utran_Cell9, U1_IDLE);
  }
}

Change 3 – Correction to ‘f_EUTRA_Preamble_Reject ()’ and ‘f_EUTRA_AttachReject ()’

Function name
f_EUTRA_Preamble_Reject()
f_EUTRA_AttachReject()

Reason for change
See change 1.2.

Summary of change
See change 1.2

TTCN module
EUTRA_NASCommonFunctions.ttcn

MCC160 Comment


Before change

  /*
   * @desc      Generic test step to reject a registration attempt by the UE.
   *            This is based on the Preamble function and should be run as an alternative.
   *            This results in the UE being switched off, and  with no valid GUTI
   * @param     p_CellId
   * @param     p_RejectCause
   * @param     p_ForcedAttach
   * @status    APPROVED
   */
  function f_EUTRA_Preamble_Reject(CellId_Type p_CellId,
                                   NAS_CauseValue_Type p_RejectCause,
                                   EUTRA_ATTACH_Type p_ForcedAttach) runs on EUTRA_PTC
  {
    // First switch cell back on
    f_EUTRA_SetCellPower ( p_CellId, tsc_ServingCellRS_EPRE );
    // Activate DHCP and ICMPv6
    f_IP_Handling_Start(IP, PDN_1, cs_DrbInfo_EUTRA(p_CellId, tsc_DRB1));    /* @sic R5s110449, R5-113734 change 8 sic@ */
    f_EUTRA_AttachReject (p_CellId, p_RejectCause, p_ForcedAttach); // @sic R5s110176 sic@
    // And then just need to switch UE off
    f_UT_SwitchOffUE (UT, false, CNF_REQUIRED); // @sic R5s1100395 sic@
  }
  /*
   * @desc      Generic test step to reject a registration attempt by the UE.
   *            This expects the UE to be switched off before calling this function, and leaves the UE switched on at the end
   * @param     p_CellId
   * @param     p_RejectCause
   * @param     p_ForcedAttach
   * @status    APPROVED
   */
  function f_EUTRA_AttachReject (CellId_Type p_CellId,
                                 NAS_CauseValue_Type p_RejectCause,
                                 EUTRA_ATTACH_Type p_ForcedAttach) runs on EUTRA_PTC
  {
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var NAS_AttDetValue_Type v_AttachType;
    var PdnTypeValue v_PDNValue;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var EUTRA_SecurityParams_Type v_AuthParams := f_EUTRA_Security_Get();
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (p_ForcedAttach); // @sic R5s110176 sic@
    // Switch on UE
    f_UT_SwitchOnUE (UT, false);
    v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5s110176 sic@
    v_PDNValue := f_GetPdnType();
    // RRC Connection Establishment, with Attach Request and PDN Connectivity
    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,
                                                v_RRC_TI,
                                                cr_EstablishmentCause_Any,
                                                cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection,
                                                                                   tsc_SHT_IntegrityProtected),
                                                                                  cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType), // @sic R5-103681 sic@
                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,
                                                                                                                  v_PDNValue)));
    // Set UE network capabilities as sent by the UE to be used later in the test case(see TS 24.301 clause 5.4.3.3)
    v_AuthParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability(v_NAS_Ind.Pdu.Msg.aTTACH_REQUEST);  // @sic R5s100135 sic@
    f_EUTRA_Security_Set(v_AuthParams);
    // Now send Attach Reject, to get UE to delete GUTI
    SRB.send  ( cas_SRB1_NasPdu_REQ ( p_CellId,
                                      cs_TimingInfo_Now,
                                      cs_NAS_Request ( tsc_SHT_NoSecurityProtection,
                                                       cs_508_AttachReject ( cs_EMM_Cause_v(p_RejectCause),
                                                                             omit))));
    f_EUTRA_RRC_ConnectionRelease(p_CellId);
  }


After change

  /*
   * @desc      Generic test step to reject a registration attempt by the UE.
   *            This is based on the Preamble function and should be run as an alternative.
   *            This results in the UE being switched off, and  with no valid GUTI
   * @param     p_CellId
   * @param     p_RejectCause
   * @param     p_ForcedAttach
   * @status    APPROVED
   */
  function f_EUTRA_Preamble_Reject(CellId_Type p_CellId,
                                   NAS_CauseValue_Type p_RejectCause,
                                   EUTRA_ATTACH_Type p_ForcedAttach,
                                   integer p_SelectedPlmnId := 1) runs on EUTRA_PTC
  {
    // First switch cell back on
    f_EUTRA_SetCellPower ( p_CellId, tsc_ServingCellRS_EPRE );
    // Activate DHCP and ICMPv6
    f_IP_Handling_Start(IP, PDN_1, cs_DrbInfo_EUTRA(p_CellId, tsc_DRB1));    /* @sic R5s110449, R5-113734 change 8 sic@ */
    f_EUTRA_AttachReject (p_CellId, p_RejectCause, p_ForcedAttach, p_SelectedPlmnId); // @sic R5s110176 sic@
    // And then just need to switch UE off
    f_UT_SwitchOffUE (UT, false, CNF_REQUIRED); // @sic R5s1100395 sic@
  }
  /*
   * @desc      Generic test step to reject a registration attempt by the UE.
   *            This expects the UE to be switched off before calling this function, and leaves the UE switched on at the end
   * @param     p_CellId
   * @param     p_RejectCause
   * @param     p_ForcedAttach
   * @status    APPROVED
   */
  function f_EUTRA_AttachReject (CellId_Type p_CellId,
                                 NAS_CauseValue_Type p_RejectCause,
                                 EUTRA_ATTACH_Type p_ForcedAttach,
                                 integer p_SelectedPlmnId := 1) runs on EUTRA_PTC
  {
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var NAS_AttDetValue_Type v_AttachType;
    var PdnTypeValue v_PDNValue;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var EUTRA_SecurityParams_Type v_AuthParams := f_EUTRA_Security_Get();
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (p_ForcedAttach); // @sic R5s110176 sic@
    // Switch on UE
    f_UT_SwitchOnUE (UT, false);
    v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5s110176 sic@
    v_PDNValue := f_GetPdnType();
    // RRC Connection Establishment, with Attach Request and PDN Connectivity
    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,
                                                v_RRC_TI,
                                                cr_EstablishmentCause_Any,
                                                cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection,
                                                                                   tsc_SHT_IntegrityProtected),
                                                                                  cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType), // @sic R5-103681 sic@
                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,
                                                                                                                  v_PDNValue)),
                                                p_SelectedPlmnId);
    // Set UE network capabilities as sent by the UE to be used later in the test case(see TS 24.301 clause 5.4.3.3)
    v_AuthParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability(v_NAS_Ind.Pdu.Msg.aTTACH_REQUEST);  // @sic R5s100135 sic@
    f_EUTRA_Security_Set(v_AuthParams);
    // Now send Attach Reject, to get UE to delete GUTI
    SRB.send  ( cas_SRB1_NasPdu_REQ ( p_CellId,
                                      cs_TimingInfo_Now,
                                      cs_NAS_Request ( tsc_SHT_NoSecurityProtection,
                                                       cs_508_AttachReject ( cs_EMM_Cause_v(p_RejectCause),
                                                                             omit))));
    f_EUTRA_RRC_ConnectionRelease(p_CellId);
  }


Change 4 – Introduction of decode function

Function name
f_DecodeRrcDataInd_Msg()

Reason for change
See change 2.1.  

Summary of change
The UTRAN_TypeDefs module is imported to include the definition of the RRCDataIndType into TTCN-3 types.

Moreover, the new function ‘f_DecodeRrcDataInd_Msg’ is defined to decode type RRCDataIndType.

TTCN module
UTRANGERAN_CommonFunctions.ttcn

MCC160 Comment


Before change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2011, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TTA, TTC).
//   All rights reserved.
// @version: IWD_11wk37
// $Date: 2011-09-09 18:47:39 +0200 (Fri, 09 Sep 2011) $
// $Rev: 5494 $
/******************************************************************************/
module UTRANGERAN_CommonFunctions {
  import from UTRAN_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from CommonDefs all;
  import from NAS_24008TypeDefs all;
  import from NAS_CommonTypeDefs all;
  import from NAS_24008Templates all;
  import from Parameters all;
//  import from NAS_AuxiliaryDefsAndFunctions all;
  import from CommonIP all;
  //----------------------------------------------------------------------------
  type enumerated      UTRAN_GERAN_Type {UTRAN_GERAN, UTRAN_Only};
  type enumerated      UTRAN_GERAN_ATTACH_Type {Combined, PS_Only, NotSpecified};  /* @status    APPROVED */
  //----------------------------------------------------------------------------
  … 

  /*
   * @desc      Function to decode the Routing Area Update Request message in to type ROUTINGAREAUPDATEREQUEST
   * @param     p_Octetstring
   * @return    ROUTINGAREAUPDATEREQUEST
   * @status
   */
  function f_DecodeRauReq_Msg ( octetstring p_Octetstring ) return ROUTINGAREAUPDATEREQUEST {
    var ROUTINGAREAUPDATEREQUEST v_ReceivedRauReq_Msg;
    var bitstring v_Bitstring := oct2bit ( p_Octetstring );
    var integer v_Result := decvalue ( v_Bitstring, v_ReceivedRauReq_Msg );
    if (v_Result != 0) {
      FatalError(__FILE__, __LINE__, "Routing Area Request message cannot be decoded");
    }
    return v_ReceivedRauReq_Msg;
  }


After Change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2011, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TTA, TTC).
//   All rights reserved.
// @version: IWD_11wk37
// $Date: 2011-09-09 18:47:39 +0200 (Fri, 09 Sep 2011) $
// $Rev: 5494 $
/******************************************************************************/
module UTRANGERAN_CommonFunctions {
  import from UTRAN_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from CommonDefs all;
  import from NAS_24008TypeDefs all;
  import from NAS_CommonTypeDefs all;
  import from NAS_24008Templates all;
  import from Parameters all;
//  import from NAS_AuxiliaryDefsAndFunctions all;
  import from CommonIP all;
  import from UTRAN_TypeDefs all; 
  //----------------------------------------------------------------------------
  type enumerated      UTRAN_GERAN_Type {UTRAN_GERAN, UTRAN_Only};
  type enumerated      UTRAN_GERAN_ATTACH_Type {Combined, PS_Only, NotSpecified};  /* @status    APPROVED */
  //----------------------------------------------------------------------------
  … 

  /*
   * @desc      Function to decode the Routing Area Update Request message in to type ROUTINGAREAUPDATEREQUEST
   * @param     p_Octetstring
   * @return    ROUTINGAREAUPDATEREQUEST
   * @status
   */
  function f_DecodeRauReq_Msg ( octetstring p_Octetstring ) return ROUTINGAREAUPDATEREQUEST {
    var ROUTINGAREAUPDATEREQUEST v_ReceivedRauReq_Msg;
    var bitstring v_Bitstring := oct2bit ( p_Octetstring );
    var integer v_Result := decvalue ( v_Bitstring, v_ReceivedRauReq_Msg );
    if (v_Result != 0) {
      FatalError(__FILE__, __LINE__, "Routing Area Request message cannot be decoded");
    }
    return v_ReceivedRauReq_Msg;
  }
  function f_DecodeRrcDataInd_Msg ( octetstring p_Octetstring ) return RRCDataIndType {
    var RRCDataIndType v_ReceivedrrcDataInd_Msg;
    var bitstring v_Bitstring := oct2bit ( p_Octetstring );
    var integer v_Result := decvalue ( v_Bitstring, v_ReceivedrrcDataInd_Msg );
    if (v_Result != 0) {
      FatalError(__FILE__, __LINE__, "Routing Area Request message cannot be decoded");
    }
    return v_ReceivedrrcDataInd_Msg;
  }

Change 5 – Correction to ‘cs_TrLogMappingRACH_DTCH’

Template name
cs_TrLogMappingRACH_DTCH

Reason for change
Because of the incorrect template definition of ‘cs_TrLogMappingRACH_DTCH’,  function ‘f_UTRAN_SS_SetConfigRRC_RB3’ reconfigures RB3 instead of RB2.

Summary of change
Template ‘cs_TrLogMappingRACH_DTCH’ is corrected to allow reconfiguration of RB3 instead of RB2.

TTCN module
UTRAN_ASN1_ASP_Templates.ttcn

MCC160 Comment


Before change

  template (value) TrCH_LogCHMappingList1 cds_TrLogMappingRACH_DTCH_RRC_RB3 modifies cs_TrLogMappingRACH_DTCH :=
  {
    ulconnectedTrCHList := {
      [0] := {
        trCH_LogCHMappingList := {
          [2] := {
            rB_Identity := tsc_RB3_DCCH_RRC
          }
        }
      }
    }
  };

After Change

  template (value) TrCH_LogCHMappingList1 cds_TrLogMappingRACH_DTCH_RRC_RB3 modifies cs_TrLogMappingRACH_DTCH :=
  {
    ulconnectedTrCHList := {
      [0] := {
        trCH_LogCHMappingList := {
          [3] := {
            rB_Identity := tsc_RB3_DCCH_RRC
          }
        }
      }
    }
};

Change 6 – Correction to ‘cr_InitDirectTransfer’

Function name
cr_InitDirectTransfer ()

Reason for change
(P)TMSI is allocated in another PLMN (i.e. GUTI allocated in another PLMN). Therefore the routing basis needs to be modified.

Summary of change
Routing field in template ‘cr_InitDirectTransfer’ has been updated from routing basis ‘localPTMSI’ to ‘tMSIofdifferentPLMN’.

TTCN module
UTRAN_RRC_Templates.ttcn

MCC160 Comment


Before change

  //================================================================================================
  // Initial Direct Transfer
  //------------------------------------------------------------------------------------------------
  template UL_DCCH_Message cr_InitDirectTransfer ( template RoutingParameter p_RoutParameter,
                                                   PLMN_Identity p_PLMN_Identity,
                                                   CN_DomainIdentity p_CN_DomainId := ps_domain,
                                                   template EstablishmentCause p_EstCause := *) :=
  {
    integrityCheckInfo := *,
    message_ := {
      initialDirectTransfer := {
        cn_DomainIdentity := p_CN_DomainId,
        intraDomainNasNodeSelector := {
          version := {
            release99 := {
              cn_Type := {
                gsm_Map_IDNNS := {
                  routingbasis := {
                    localPTMSI := {
                      routingparameter := p_RoutParameter
                    }
                  },
                  dummy := ?
                }
              }
            }
          }
        },
        nas_Message := ?,
        measuredResultsOnRACH := *,
        v3a0NonCriticalExtensions := {
          initialDirectTransfer_v3a0ext := {
            start_Value := '00000000000000000000'B
          },
          laterNonCriticalExtensions := {
            initialDirectTransfer_r3_add_ext := *,
            v590NonCriticalExtensions := {
              initialDirectTransfer_v590ext := {
                establishmentCause := p_EstCause
              },
              v690NonCriticalExtensions := {
                initialDirectTransfer_v690ext := {
                  plmn_Identity := p_PLMN_Identity,
                  measuredResultsOnRACHinterFreq := *,
                  mbms_JoinedInformation := *
                },
                v770NonCriticalExtensions := *
              }
            }
          }
        }
      }
    }
  };

After Change

  //================================================================================================
  // Initial Direct Transfer
  //------------------------------------------------------------------------------------------------
  template UL_DCCH_Message cr_InitDirectTransfer ( template RoutingParameter p_RoutParameter,
                                                   PLMN_Identity p_PLMN_Identity,
                                                   CN_DomainIdentity p_CN_DomainId := ps_domain,
                                                   template EstablishmentCause p_EstCause := *) :=
  {
    integrityCheckInfo := *,
    message_ := {
      initialDirectTransfer := {
        cn_DomainIdentity := p_CN_DomainId,
        intraDomainNasNodeSelector := {
          version := {
            release99 := {
              cn_Type := {
                gsm_Map_IDNNS := {
                  routingbasis := {
                    tMSIofdifferentPLMN := {
                      routingparameter := p_RoutParameter
                    }
                  },
                  dummy := ?
                }
              }
            }
          }
        },
        nas_Message := ?,
        measuredResultsOnRACH := *,
        v3a0NonCriticalExtensions := {
          initialDirectTransfer_v3a0ext := {
            start_Value := '00000000000000000000'B
          },
          laterNonCriticalExtensions := {
            initialDirectTransfer_r3_add_ext := *,
            v590NonCriticalExtensions := {
              initialDirectTransfer_v590ext := {
                establishmentCause := p_EstCause
              },
              v690NonCriticalExtensions := {
                initialDirectTransfer_v690ext := {
                  plmn_Identity := p_PLMN_Identity,
                  measuredResultsOnRACHinterFreq := *,
                  mbms_JoinedInformation := *
                },
                v770NonCriticalExtensions := *
              }
            }
          }
        }
      }
    }
  };

5. Execution Log Files

5.1 Qualcomm MDM 9200 UE 

The Qualcomm MDM 9200 UE passed this test case on R&S CMW500 with NAS and AS Snow 3G integrity and ciphering algorithm in LTE FDD band 1 and UTRA FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
tc_6_2_1_3_Qualcomm_FDD_band_1.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

References

[1]
R5s110691: Supporting information for agreement of TC 6.2.1.3. This archive comprises:

- text format execution log files

